

Capítulo 12

Pantalla de selección



For the presentation of this unit, you should know the sequence of the ABAP events such as *INITIALIZATION*, *START-OF-SELECTION* and so on.

Resumen del capítulo

En esta unidad, se presentan los comandos ABAP para generar una pantalla de selección, así como también otras adiciones y técnicas. Se analizan los eventos ABAP que pertenecen a la pantalla de selección (como *AT SELECTION-SCREEN*). La creación de variantes también se trata así como la modificación dinámica de pantallas de selección.



Objetivos del capítulo

Al finalizar este capítulo podrá:

- Explicar el objetivo de las pantallas de selección
- Crear campos de entrada con *PARAMETERS*
- Crear campos de entrada con *SELECT-OPTIONS*
- Diseñar la pantalla de selección
- Definir pantallas de selección adicionales
- Definir subscreens
- Definir una ficha en una pantalla de selección
- Implementar verificaciones de entrada para las pantallas de selección
- Crear variantes para una pantalla de selección
 - Crear pulsadores en la pantalla de selección
- Programar cambios en la pantalla de selección en el tiempo de ejecución

Lección: Generación y diseño de la pantalla de selección

Resumen de la lección

En esta lección aprenderá cómo crear y diseñar una pantalla de selección y cómo verificar las entradas. Finalmente, aprenderá cómo se pueden cambiar los elementos de pantalla en tiempo de ejecución.



Objetivos de la lección

Al finalizar esta lección podrá:

Explicar el objetivo de las pantallas de selección

Crear campos de entrada con PARAMETERS

Crear campos de entrada con SELECT-OPTIONS

Diseñar la pantalla de selección

Definir pantallas de selección adicionales

Definir subscreens

Definir una ficha en una pantalla de selección

Ejemplo empresarial

Usted es programador en ABC Travel. Su gerente desea ver los datos de vuelo para ABC Airways. Su tarea es escribir un programa con una pantalla de selección fácil de usar para cumplir con los requisitos del gerente.

Resumen de la pantalla de selección




Gráfico 236: Pantalla de selección: resumen

Las pantallas de selección funcionan como la interfaz entre el programa y el usuario, y se utilizan para limitar la cantidad de datos que se leerán desde la base de datos.

Puede usar los elementos lingüísticos declarativos `PARAMETERS` y `SELECT-OPTIONS` para generar una pantalla de selección por defecto (pantalla 1000) con campos listos para la entrada.

Las bases de datos lógicas proporcionan pantallas de selección cuya apariencia concreta depende del nombre del nodo especificado (`NODES <name>`). (Consulte la unidad “Bases de datos lógicas” para obtener más detalles). Las versiones de la pantalla de selección ofrecen un subconjunto de la pantalla de selección por defecto.

Además de la pantalla de selección por defecto, puede crear pantallas de selección adicionales usando `SELECTION-SCREEN BEGIN OF ...` y llamarlas mediante `CALL SELECTION-SCREEN`.

Cree variantes para grabar valores de la pantalla de selección que se utilizan repetidamente. La variante puede recuperarse para utilizarse cuando sea necesaria. Si un informe se ejecuta con una pantalla de selección en el proceso de fondo, se requiere una variante.

Creación de campos en la pantalla de selección



```
PARAMETERS <f>
[TYPE <type>] [LIKE <f1>] [DECIMALS <n>] [MEMORY ID <pid>]
[OBLIGATORY] [DEFAULT <value>] [LOWER CASE] [VALUE CHECK]
[AS CHECKBOX]
[RADIOBUTTON GROUP <grp>] [MODIF ID <mod>].
```

```
REPORT sapbc405_sscd_checkbox_radiobutton.
...
PARAMETERS: pa_carr TYPE sflight-carrid,
             pa_name AS CHECKBOX DEFAULT 'X',
             pa_curr AS CHECKBOX DEFAULT 'X',
             pa_lim_1 RADIOBUTTON GROUP lim,
             pa_lim_2 RADIOBUTTON GROUP lim,
             pa_lim_3 RADIOBUTTON GROUP lim.
CONSTANTS mark VALUE 'X'.

* Check if any checkbox has been selected
IF pa_name EQ mark. ... . ENDIF.
IF pa_curr EQ mark. ... . ENDIF.

* Check which radiobutton has been selected
CASE mark.
  WHEN pa_lim_1. ... .
  WHEN pa_lim_2. ... .
  WHEN pa_lim_3. ... .
ENDCASE.
```

Cía. aérea AA

☒ Nombre de salida

☒ Moneda local de salida

☒ Precio (moneda local): hasta 500

☐ Precio (moneda local): desde 501 hasta 1000

☐ Precio (moneda local): más de 1000

Gráfico 237: Declaración de campos con PARAMETERS

La sentencia `PARAMETERS` es un elemento lingüístico declarativo. Como en el caso de la sentencia `DATA`, puede declarar parámetros con `TYPE` o `LIKE`. El sistema genera campos listos para la entrada en la pantalla de selección. Los nombres de los campos `PARAMETERS` pueden tener una longitud de hasta ocho caracteres. Puede actualizar textos de selección usando el acceso vía menús *Pasar a* → *Elementos de texto* → *Textos de selección*.

Con el suplemento `DEFAULT <wert>`, puede establecer un valor por defecto para una sentencia `PARAMETERS`. Si asigna una `MEMORY ID <pid>`, el sistema recupera el valor actual de la memoria SAP y le da salida en la pantalla automáticamente. La memoria SAP es un área de memoria específica del usuario donde se guardan valores de datos simples (por ejemplo, la clave para una aerolínea) para parámetros SET/GET mientras dure una sesión de usuario. Los parámetros SET/GET se identifican por nombre (longitud máxima 20 caracteres). Puede descubrir el nombre de un parámetro SET/GET relevante consultando la *Información técnica* en un campo de salida mediante *Ayuda(F1)* → *Información técnica*. En el campo *ID de parámetro*, encontrará el nombre del parámetro SET/GET como se propone en el elementos de datos que se está utilizando.

Si declara campos obligatorios (suplemento `OBLIGATORY`), los usuarios no pueden salir de la pantalla de selección hasta que estos campos contengan entradas.

La verificación automática realizada por el Dictionary ABAP se desactiva por defecto para las pantallas de selección de informes. Sin embargo, si hace referencia a un objeto de Dictionary que tiene una clave externa establecida, `VALUE CHECK` asegura que se realiza la verificación de la validez de las entradas.

Puede generar parámetros como casillas de selección (`AS CHECKBOX`). Como resultado, se crea un campo de un carácter que contiene (`space`) o “X”. Puede evaluar el contenido de las casillas de selección en estructuras `IF/ENDIF`.

Además, puede generar un grupo de botones de selección para la pantalla de selección usando el suplemento `RADIOBUTTON GROUP <grp>`. El nombre `<grp>` de un grupo de botones de selección puede tener un máximo de cuatro caracteres. Únicamente **un** botón de selección por grupo puede estar activo; puede evaluarse en el procesamiento de programas. Puede evaluar el contenido de los botones de selección mediante una estructura de control `CASE/ENDCASE`. El botón de selección que elige tiene el valor “X”, todos los demás tienen el valor (`space`). Si no ha asignado un valor a ninguno de los botones de selección, el sistema automáticamente seleccionará el primer botón de selección cuando muestre la pantalla de selección.

Puede usar el suplemento `MODIF ID <mod>` para simplificar las modificaciones dinámicas de la pantalla de selección (para obtener más información, consulte la unidad “Pantalla de selección: modificaciones en tiempo de ejecución”).




Gráfico 238: Selecciones con SELECT-OPTIONS

`SELECT-OPTIONS` es un elemento lingüístico declarativo con la misma restricción de caracteres (ocho caracteres) que `PARAMETERS`. A diferencia de `PARAMETERS`, `SELECT-OPTIONS` permite diferentes valores y selecciones complejas, en lugar de solamente un único campo de entrada.

`SELECT-OPTIONS` genera una tabla interna `<seltab>` con una estructura estándar y una línea de cabecera. La tabla tiene cuatro columnas: `sign` (indicador que significa incluido o sólo), `option` (operador relacional), `low` (límite inferior) y `high` (límite superior). Al igual que con `PARAMETERS`, puede actualizar los textos de selección usando el acceso vía menús *Pasar a* → *Elementos de texto* → *Textos de selección*.

Después del parámetro `FOR`, puede especificar el campo con el que el sistema debe verificar las entradas de selección. Debe declarar el campo, por ejemplo, en una sentencia `DATA` o `NODES`. Tanto el campo alto como el bajo (`high` y `low`) heredan los atributos del campo al que se hace referencia.

Todas las líneas de la tabla de selección `<seltab>` formulan una condición mediante un operador relacional. Los siguientes valores son posibles:



Posibles valores para SIGN y OPTION

	Valor	Significado
SIGN	I	Incluir
	E	Excluir
OPTION	EQ	Igual
	NE	No igual
	LE	Menos o igual
	LT	Menor que
	GE	Mayor o igual
	GT	Mayor que
	BT	Entre
	NB	No entre
	CP	Contiene patrón
	NP	Contiene patrón no

El set de selección `S` está formado por incluye todo (I_1, \dots, I_n) menos excluye todo (E_1, \dots, E_m). Si no se ingresa ningún valor en el objeto `SELECT-OPTIONS`, el sistema regresa todas las líneas de la base de datos (“Nada es todo”).




Gráfico 239: SELECT-OPTIONS con criterios de selección múltiple

Cuando el usuario realiza entradas en un objeto SELECT-OPTIONS, el sistema proporciona la tabla interna automáticamente.

Para cambiar las entradas por defecto para los campos de tabla sign y option, elija *Editar* → *Opciones de selección*. El sistema ofrece todas las alternativas apropiadas para la selección. Si el semáforo está en verde para *Seleccionar*, hay una **I** (inclusión) en la columna SIGN, mientras que la luz roja indica **E** (exclusión).

Para borrar una tabla de entrada, elija *Editar* → *Borrar criterio de selección*.

Todos los criterios de selección pueden utilizarse para realizar selecciones múltiples, a menos que se defina lo contrario. Si las selecciones múltiples están presentes, el color de la flecha cambia de blanco a verde.



SELECT-OPTIONS <seltab> FOR <f>	
DEFAULT <value>	SIGN <x> OPTION <xx>
DEFAULT <value1> TO <value2>	
MEMORY ID <pid>	
LOWER CASE	
OBLIGATORY	
NO-EXTENSION	
NO INTERVALS	
MODIF ID <mod>	

Gráfico 240: Sintaxis de la sentencia SELECT-OPTIONS

Los suplementos de la sentencia SELECT-OPTIONS son:

DEFAULT le permite establecer valores por defecto para los parámetros low o low y high. Puede utilizar OPTION y SIGN para determinar valores por defecto para option y sign que difieren de los valores por defecto.

Si asigna una MEMORY ID <pid>, el sistema recupera el valor actual de la memoria SAP, le da salida en la pantalla y lo escribe nuevamente cuando sale de la pantalla (se incluyen los cambios que pudiera haber hecho).

LOWER CASE suprime la conversión de la entrada en mayúsculas, siempre y cuando el dominio lo permita o la tipificación se haya realizado en el tipo C de ABAP.

OBLIGATORY genera un campo de entrada requerido. Aparece una marca de verificación en el campo de entrada en la pantalla de selección, y el usuario debe ingresar un valor.

NO-EXTENSION suprime la opción de selección múltiple.

NO INTERVALS suprime el límite superior del intervalo (<seltab>-high) en la pantalla de selección. Puede utilizar la pantalla adicional *Selección múltiple* para ingresar intervalos.

Asignación dinámica previa de la pantalla de selección




Gráfico 241: Eventos ABAP en el entorno de la pantalla de selección

El evento `LOAD-OF-PROGRAM` se ejecuta si el código de la máquina está cargado en la memoria principal, es decir, justo al comienzo de la ejecución del programa. Por ahora, los valores por defecto de los campos de entrada no se han evaluado.

Por lo tanto, no se recomienda la asignación dinámica previa de los campos de entrada en este momento, ya que los valores por defecto sobrescribirían las asignaciones previas.

`AT SELECTION-SCREEN OUTPUT` se ejecuta inmediatamente antes de que se visualice la pantalla de selección, mientras que `AT SELECTION-SCREEN` se ejecuta inmediatamente después de que se visualice la pantalla de selección. Como la pantalla de selección se puede visualizar varias veces (para *INTRO*, por ejemplo), el sistema también puede procesar esos bloques de evento muchas veces.



INITIALIZATION.

REPORT sapbc405_sscd_initialization.
...

INITIALIZATION.

MOVE: mark TO pa_all.

MOVE: 'I' TO so_carr-sign,
'BT' TO so_carr-option,
'AA' TO so_carr-low,
'LH' TO so_carr-high.

APPEND so_carr.
CLEAR so_carr.
MOVE: 'E' TO so_carr-sign,
'EQ' TO so_carr-option,
'DL' TO so_carr-low.

APPEND so_carr.
...

Cía. aérea a

Fecha de vuelo a

Salida ...

Asientos ...

Ocupados ☐
Libres ☐
Todos ☒

Selección Colores ☐ Iconos ☐

Gráfico 242: Inicialización de la pantalla de selección

El sistema sólo procesa el evento `INITIALIZATION` una vez en cada programa ejecutable. Durante este evento, puede proporcionar valores por defecto para los campos de la pantalla de selección de un informe, así como también aquéllos para un base de datos lógica. Puede usar Ayuda F1 (Ayuda técnica) para determinar los nombres de los campos de selección si se conecta una base de datos lógica con el programa.



```
DATA: wa_sbook TYPE sbook.
SELECT-OPTIONS: so_car FOR wa_sbook-carrid.
```

so_car (línea de cabecera)

```
MOVE: 'I' TO SO_CAR-SIGN,
      'BT' TO SO_CAR-OPTION,
      'AA' TO SO_CAR-LOW,
      'QF' TO SO_CAR-HIGH.
APPEND SO_CAR.
```

Signo	Opción	Bajo	Alto
I	BT	AA	QF

```
CLEAR SO_CAR.
MOVE: 'E' TO SO_CAR-SIGN,
      'BT' TO SO_CAR-OPTION,
      'AZ' TO SO_CAR-LOW.
APPEND SO_CAR.
```

Signo	Opción	Bajo	Alto
E	EQ	AZ	

so_car (tabla interna)

Signo	Opción	Bajo	Alto
I	BT	AA	QF
E	EQ	AZ	

APPEND so_car causas:

El contenido de la línea de cabecera so_car debe conectarse al cuerpo de la tabla so_car como una nueva línea

Gráfico 243: Relleno de las tablas internas con líneas de cabecera

La sentencia `SELECT-OPTIONS <sel_opt> FOR <data_object>` siem-pre genera una tabla interna con una línea de cabecera. En realidad, esto crea dos objetos de datos: el cuerpo de una tabla llamado `<sel_opt>` y un área de trabajo línea de cabecera), también llamada `<sel_opt>`. Rellene el área de trabajo `<sel_opt>` con valores adecuados, luego use el comando `APPEND <sel_opt>` para agregarlo al cuerpo de la tabla como una línea distinta.

En la pantalla de selección también se pueden agregar valores en el evento `AT SELECTION-SCREEN OUTPUT`. Este evento sucede justo antes de que se visualice la pantalla de selección y se puede ejecutar varias veces.

Diseño de la pantalla de selección



```

SELECTION-SCREEN BEGIN OF BLOCK <block>
WITH FRAME TITLE <text>
SELECTION-SCREEN END OF BLOCK <block>

REPORT sapbc405_sscd_sel_screen_i.
...
SELECTION-SCREEN BEGIN OF BLOCK carr WITH FRAME.
  SELECT-OPTIONS: so_carr FOR wa_sflight-carrid.
SELECTION-SCREEN END OF BLOCK carr.

SELECTION-SCREEN BEGIN OF BLOCK limit WITH FRAME TITLE text-001.
  PARAMETERS:
    pa_lim_1 RADIOBUTTON GROUP lim,
    pa_lim_2 RADIOBUTTON GROUP lim,
    pa_lim_3 RADIOBUTTON GROUP lim.
SELECTION-SCREEN END OF BLOCK limit.
...

```

Gráfico 244: Diseño de la pantalla de selección I

Puede usar la sentencia `SELECTION-SCREEN` para diseñar la presentación de la pantalla de selección.

Puede usar el suplemento `BEGIN OF BLOCK <block>` para agrupar elementos de la pantalla relacionados lógicamente y `WITH FRAME`, con el objetivo de enmarcarlos. Los marcos pueden anidarse hasta alcanzar una profundidad máxima de cinco.

Por razones de accesibilidad, debe asignarle una cabecera al bloque. El título se puede declarar como `text-xxx` o como un nombre de campo de hasta ocho caracteres. En el segundo caso, el texto se establece en tiempo de ejecución en el evento `INITIALIZATION`. Antes de diseñar una pantalla de selección, debe familiarizarse con las pautas de diseño de pantallas que se muestran en la transacción BIBS.



SELECTION-SCREEN:

BEGIN OF LINE

COMMENT pos(len) <text> [FOR FIELD <f>]

POSITION pos

END OF LINE

```
REPORT sapbc405_sscd_sel_screen_ii.
...
* Parameters displayed in a row
  SELECTION-SCREEN BEGIN OF LINE.
    SELECTION-SCREEN COMMENT 1(20) text-s03.
    SELECTION-SCREEN COMMENT pos_low(8) text-s04 FOR FIELD
                                     pa_col.

    PARAMETERS pa_col AS CHECKBOX.
    SELECTION-SCREEN COMMENT pos_high(8) text-s05.
    PARAMETERS pa_ico AS CHECKBOX.
  SELECTION-SCREEN END OF LINE.
```

Gráfico 245: Diseño de la pantalla de selección II

Puede mostrar diversos parámetros y comentarios en la misma línea de la pantalla. Para hacerlo, debe incluirlos entre SELECTION-SCREEN BEGIN OF LINE y SELECTION-SCREEN END OF LINE. El suplemento COMMENT le permite incluir textos.

Los textos comentarios siempre deben tener un formato (posición y longitud de salida). La posición se puede determinar usando un campo de datos o pos_low, o bien pos_high. Los últimos dos son las posiciones para los límites superiores e inferiores del campo SELECT-OPTIONS en la pantalla de selección.

El suplemento <COMMENT ... FOR FIELD <f> hace que el sistema automáticamente muestre la ayuda F1 del campo <f> para el texto comentario y el parámetro. Si oculta el parámetro (variante de selección: atributo invisible), el texto de comentario también se oculta.

Puede usar POSITION <pos> para ubicar el cursor en una línea para la próxima salida (sólo en BEGIN OF LINE ... END OF LINE).

Etiquetas de la pantalla de selección

Ahora puede visualizar controles de fichas en las pantallas de selección. Le permiten crear grupos de campos lógicos y facilitan el uso de las pantallas de selección grandes.



- Disposición concisa de imágenes de selección grandes
- Agrupaciones de campos lógicos




Gráfico 247: Controles de fichas en la pantalla de selección

Para una pantalla de selección con controles de fichas, debe definir:

Un área subscreen en la pantalla de selección para ajustar el control de fichas

Los títulos de las etiquetas individuales

Pantallas de selección como subscreens para los títulos de las etiquetas individuales.



```
DATA: wa_spfli TYPE spfli.

SELECTION-SCREEN BEGIN OF SCREEN 101 AS SUBSCREEN.
  SELECT-OPTIONS: so_car FOR wa_spfli-carrid,
                  so_con FOR wa_spfli-connid.
SELECTION-SCREEN END OF SCREEN 101.

SELECTION-SCREEN BEGIN OF SCREEN 102 AS SUBSCREEN.
  PARAMETERS:    ...
SELECTION-SCREEN END OF SCREEN 102.

SELECTION-SCREEN BEGIN OF SCREEN 103 AS SUBSCREEN.
  PARAMETERS:    ...
SELECTION-SCREEN END OF SCREEN 103.
```

Subscreen 101



Subscreen 102



Subscreen 103

**Gráfico 248: Pantallas de selección como Subscreens**

Como las pantallas de selección pueden definirse como subscreens, puede incluirlas en cualquier otra pantalla, especialmente en etiquetas. El sistema procesa los eventos ABAP AT SELECTION-SCREEN OUTPUT y AT SELECTION-SCREEN para cada subscreen, además de procesar la pantalla de selección circundante. Utilice las variables del sistema sy-dynnr para determinar qué pantalla (pantalla de selección o subscreen) se está procesando actualmente.

Una pantalla de selección puede procesarse como una subscreen de la siguiente manera:

```
SELECTION-SCREEN BEGIN OF SCREEN <dynnr> AS
SUBSCREEN [NESTING LEVEL <m>] [NO INTERVALS].
....
SELECTION-SCREEN END OF SCREEN <dynnr>.
```

Suplementos opcionales:

[NESTING LEVEL <m>]: el nivel de anidamiento NESTING

LEVEL reduce aun más el tamaño de la subscreen. Puede utilizarlo para evitar que las barras de desplazamiento aparezcan cuando use la subscreen en un control de fichas en la pantalla de selección, y la ficha ya tiene un marco. Si el control de fichas no tiene un marco, use NESTING LEVEL 0. Para cada marco que se encuentre alrededor del control de fichas, aumente NESTING LEVEL en 1.

[NO INTERVALS] Esta opción oculta los campos HIGH para todos los criterios de selección definidos que usan SELECT-OPTIONS en la pantalla.




Gráfico 249: Definición de fichas en la pantalla de selección

Una subscreen para un control de fichas se define en una pantalla de selección de la siguiente manera:

```
SELECTION-SCREEN BEGIN OF TABBED BLOCK
<blockname> FOR <n> LINES.
SELECTION-SCREEN END OF BLOCK <blockname>
```

La altura del área subscreen en líneas se define con <n>.

Las etiquetas individuales se definen de la siguiente manera:

```
SELECTION-SCREEN TAB (max_length) <name> USER-COMMAND
<ucomm> [DEFAULT
[PROGRAM <prog>] SCREEN <dynnr>].
```

Si usa el suplemento DEFAULT, también debe usar el suplemento SCREEN. El suplemento PROGRAM es opcional. Sólo lo necesitará si la pantalla procede de otro programa.

Puede posponer la especificación del enlace entre el título de la etiqueta y la pantalla de selección hasta el tiempo de ejecución. También puede modificar una asignación existente en tiempo de ejecución. Para ello, rellene la estructura <blockname>. Ésta se crea de forma automática para cada bloque de fichas. La estructura tiene el mismo nombre que el bloque de fichas y contiene los campos prog, dynnr y activetab.

Elija nombres de variable para *tab1*, *tab2* y *tab3*. Estas variables no se declaran individualmente porque forman parte del control de fichas.

Para el evento AT SELECTION-SCREEN, la variable de sistema *sy-ucomm* contiene el comando de usuario declarado para la etiqueta actualmente seleccionada (aquí, *COMM1*, *COMM2* o *COMM3*).

Generación de pulsadores en la pantalla de selección



```
* Selection screen structure with user command
TABLES sscrfields.

* Pushbutton
SELECTION-SCREEN
    PUSHBUTTON /pos_low(12) det_on USER-COMMAND on.

* Pushbutton text
INITIALIZATION.
    det_on = 'Show details' (s01).
```

Gráfico 253: Generación de pulsadores en la pantalla de selección

Puede generar pulsadores en la pantalla de selección. Use la declaración `SELECTION-SCREEN PUSHBUTTON` para hacerlo. Tenga en cuenta que, por defecto, el pulsador no se ubica en una línea nueva. Para tener un salto de línea explícito, combine su especificación de posición con una barra diagonal, por ejemplo `/1(20)`.

Se deben especificar un nombre y un comando de usuario para el pulsador.

El texto que debe aparecer en el pulsador debe actualizarse de forma separada. Para hacerlo, se utilizan símbolos de texto que se asignan a `INITIALIZATION` del pulsador. De forma alternativa, también se puede utilizar el módulo de funciones `ICON_CREATE` para generar un texto o un ícono.

Si un usuario selecciona un pulsador, se desencadenará el comando de usuario asignado a éste y se ubicará en el campo `sscrfields-ucomm`. Por lo tanto, se debe incluir la declaración `TABLES sscrfields` en el programa.




Gráfico 254: Eventos ABAP para la pantalla de selección frente a Eventos de la pantalla

Las pantallas de selección son pantallas especiales generadas por el sistema que sólo se deben a declaraciones ABAP. En una programación de pantalla normal, para cada pantalla, existen dos puntos en el tiempo en los que se puede implementar la lógica de flujo.

Proceso antes de la salida (PBO) inmediatamente antes de la visualización de pantalla

Proceso después de la salida (PAI) inmediatamente después de la visualización de pantalla

Los equivalentes para estos eventos en la pantalla de selección son:

Proceso antes de salida (PBO): AT SELECTION-SCREEN OUTPUT

Proceso después de la entrada (PAI): AT SELECTION-SCREEN




Gráfico 255: Evaluación de pulsadores en la pantalla de selección

El comando de usuario se fija en el campo `sscrfields-ucomm` en el PAI de la pantalla de selección. El PAI se asigna mediante el evento `AT SELECTION-SCREEN`.

Por lo tanto, puede determinar qué pulsador seleccionó el usuario y reaccionar de acuerdo a esto en la aplicación. Lo ideal sería observarlo a causa de una distinción de casos en una declaración `CASE`.

Modificación de la pantalla de selección en el tiempo de ejecución




Gráfico 256: Visualización y ocultamiento de los campos de selección

Cada campo de la pantalla de selección tiene una entrada en la tabla del sistema SCREEN. Puede cambiar los contenidos de esta tabla.

Puede ocultar los campos de selección asignando el valor 0 al atributo SCREEN-ACTIVE. Puede controlar qué campos se deben ocultar usando el nombre de campo o el nombre de grupo.

Para la preparación para los status de entrada de los campos, debe asignar un valor al campo screen-input.



```
DATA: wa_sbook TYPE sbook.
SELECT-OPTIONS:
  so_car FOR wa_sbook-carrid,
  so_con FOR wa_sbook-connid MODIF ID abc.

* Screen modification
AT SELECTION-SCREEN OUTPUT.

LOOP AT SCREEN.
  IF screen-name = 'SO_CAR-LOW'
    OR screen-group1 = 'ABC'.
    screen-active = 0.
    MODIFY SCREEN.
  ENDIF.
ENDLOOP.
```

SCREEN-NAME	SO_CON
SCREEN-GROUP1	ABC
SCREEN-ACTIVE	1. 0
SCREEN-INPUT	1. 0

Gráfico 257: LOOP EN LA PANTALLA

Cada campo en la pantalla de selección tiene un nombre, por ejemplo, so_car-low y so_car-high para low o el campo high de SELECT-OPTIONS so_car.

Puede dirigirse a estos campos dentro de LOOP AT SCREEN ... ENDLOOP usando sus nombres (campo screen-name) y visualizarlos u ocultarlos en la pantalla de selección en el tiempo de ejecución. Deben visualizarse u ocultarse en el PBO de la pantalla de selección. Éste se asigna mediante el evento AT SELECTION-SCREEN OUTPUT.

Para no tener la necesidad de dirigirse a cada campo individual, puede combinar las selecciones y crear los denominados grupos de modificación. Se asigna el nombre de un grupo de modificación a la declaración PARAMETERS o SELECT-OPTIONS con el suplemento MODIF ID <xxx>.